
Beachten Sie unsere Hinweise und Nutzungsbedingungen. Vorgestellte Musterlösungen basieren auf 

Eine Übersicht zu unseren Excel-Informationen finden Sie hier: www.urs-beratung.de/toolbox.htm  
 
 
Excel - Thema: SUMMENPRODUKT 

 
SUMMENPRODUKT  
 
 

Einleitung 

Erste Erläuterungsversuche 

In der Excel-Hilfe lesen wir:  
 

„SUMMENPRODUKT multipliziert die einander entsprechenden Komponenten der angegebenen 
Matrizen miteinander und gibt die Summe dieser Produkte zurück.“  

 
Aha? – Mit unseren eigenen Worten haben wir jetzt vermutlich auch (noch) nicht mehr Erfolg:  
 

SUMMENPRODUKT ist eine Array-Funktion / Matrix-Funktion; sie multipliziert die Inhalte mehrerer 
Spalten (oder Zeilen) und addiert die Produkte zu einer Summe.  

 
Das half gewiss auch noch nicht.  
 
Versuchen wir es mit einem Beispiel zum warmwerden:  
 

1
2
3
4
5
6
7
8

A B C D E F G H
Menge Preis Umsatz

Prod.1 Prod.2 Prod.3 Prod.1 Prod.2 Prod.3 Prod.1,2,3
1 3 5 10 20 30 220 =SUMMENPRODUKT(A3:C3;D3:F3)

10 =A3*D3
60 =B3*E3

150 =C3*F3
220 =SUMME(G5:G7)  

 
G3:  
Die Absatzmengen für die drei Artikel (hier in Zellen A3 bis C3) werden mit ihren jeweiligen Preisen 
(Zellen D3 bis F3) multipliziert und zu einer gemeinsamen Summe addiert.  
 
In den Zeilen 5 bis 8 ein Erklärungsversuch dafür, wie die Funktion SUMMENPRODUKT arbeitet. 

 

http://www.urs-beratung.de/toolbox.htm#Hinweise
http://www.urs-beratung.de/impressum.htm
mailto:urs.toolbox@urs-beratung.de
http://www.urs-beratung.de/toolbox.htm

Tor zur Matrixwelt  

Wenn es so etwas wie einen typischen Anwendungsfall für die Funktion SUMMENPRODUKT gibt, dann 
sehen Sie ihn wohl oben. Dennoch: Obwohl dieses Beispiel zum warmwerden das Grundprinzip von 
SUMMENPRODUKT recht treffend beschreiben mag – die dem Beispiel zugrunde liegende simple 
Aufgabenstellung wird weder die Mehrzahl der Fälle darstellen, in denen SUMMENPRODUKT zum Einsatz 
kommt, noch wird es der Leistungsfähigkeit dieser Funktion auch nur ansatzweise gerecht.  
 
Mit unseren Erläuterungen zu den Funktionen wie SVERWEIS, INDEX, VERGLEICH hatten wir mit Ihnen 
bereits das Tor zur Matrixwelt durchschritten.  
 

 
Mit SUMMENPRODUKT machen wir nun einen tieferen Schritt 

in die Matrixwelt 
- auf der anderen Seite sieht es ein wenig anders aus.  

 

 
Erwarten Sie von diesem Papier bitte keine allumfassende Erläuterung zu Matrixformeln – unser Ziel ist 
vorrangig, die Funktion SUMMENPRODUKT zu erläutern, und auch das nur in ihren Grundzügen!  
 
 

Definition der Funktion SUMMENPRODUKT 

Definit ion  

Die Funktion erwartet lediglich die Angabe der Matrizen:  
 

=SUMMENPRODUKT(Matrix1; Matrix2; … ; Matrix30) 
 
Bis zu 30 Matrizen können in der Funktion genutzt werden. Genauer wäre von bis zu 30 Ergebnismatrizen 
zu sprechen. Warum das? Nehmen wir die erste Formel aus dem Beispiel zum warmwerden:  
 

=SUMMENPRODUKT( A3:C3 ; D3:F3 ) 
 

Hier haben wir zwei Matrizen, ordentlich durch Semikolon getrennt: Die erste Matrix betrifft die 
Zellen A3:C3, die zweite Matrix die Zellen D3:F3.  

 
Nun liefert folgende Funktion (zu unserem Beispiel oben) exakt das gleiche Ergebnis: 
 

=SUMMENPRODUKT( A3:C3 * D3:F3 ) 
 

Und hier haben wir nun nicht zwei, sondern nur eine Matrix! Durch die Multiplikation erzeugen wir 
das, was wir oben als Ergebnismatrix bezeichneten. Aus den Produkten der Zellen A3*D3 und 
B3*E3 und C3*F3 entsteht hier eine Matrix mit den Werten { 10; 60; 150 }. – Das können Sie 
nachvollziehen, indem Sie alle Zeichen innerhalb der SUMMENPRODUKT-Formel markieren und F9 
drücken (verlassen Sie die Zelle danach durch drücken der F9-Taste, ansonsten stehen die 
Ergebnisse dieser Matrix als Text in Ihrer Formel). 
 
Was die Begrenzung auf 30 Matrizen angeht – hier haben wir noch 29 frei, in der Variante mit 
Semikolon nur noch 28.  

 
 
Auf dieses Thema gehen wir unten („Semikolon oder Multiplikation?“) nochmals ein.  

Keine ganzen Spalten /  Zeilen 

SUMMENPRODUKT kann – wie auch die individuell erstellten Matrixformeln (erkennbar an den 
geschweiften { } Klammern) –  NICHT mit ganzen Spalten und ganzen Zeilen arbeiten.  
 
Es bedarf also stets einer konkreten Zellenbegrenzung.  
 

Zulässig:  =SUMMENPRODUKT(A1:B10*C1:D10) 
Nicht zulässig:  =SUMMENPRODUKT(A:B*C:D) 

 

Performance-Probleme  

SUMMENPRODUKT berechnet – ebenfalls wie individuell erstellte Matrixformeln – stets den gesamten 
angegebenen Zellenbereich. Das kann die Performance ganz erheblich beeinflussen, also Ihren PC über 
viele Minuten beschäftigen oder ihn gar zum Absturz bringen!  
 

=SUMMENPRODUKT(A3:C3;D3:F3) 
Eine Formel wie vorstehende verursacht vier Berechnungen (incl. Summation).  
 
=SUMMENPRODUKT(A1:C65536;D1:F65536) 
Hier werden dagegen schon 196.609 Berechnungen ausgeführt!  
Das gilt auch dann, wenn überhaupt nur in Zeile 3 Werte enthalten sind.  

 
Der zu berechnende Bereich ist also mit bedacht zu wählen – mit anderen Funktionen kombiniert lässt sich 
der Bereich ggf. erheblich begrenzen.   
 
 

Zum Verständnis 

Zum besseren Verständnis zeigen wir Ihnen einige Beispiele für SUMMENPRODUKT. Welchen Rechenweg 
die Funktion SUMMENPRODUKT dabei beschreitet zeigen wir jeweils in den per-pedes-Berechnungen 
darunter. 

Matrix 1 

1
2
3
4
5

A B C D E F
1 1 1 225 =SUMMENPRODUKT(A1:A5*B1:B5*C1:C5)
2 2 2 225 =A1*B1*C1 + A2*B2*C2 + A3*B3*C3 + A4*B4*C4 + A5*B5*C5
3 3 3
4 4 4 360 =SUMMENPRODUKT(A1:C1*A2:C2*A3:C3*A4:C4*A5:C5)
5 5 5 360 =A1*A2*A3*A4*A5 + B1*B2*B3*B4*B5 + C1*C2*C3*C4*C5  

 
Sie haben das Beispiel zum Warmwerden verstanden. Also ist das hier eine Kleinigkeit für Sie. 

Matrix 2 

1
2
3
4
5

A B C D E F
1 1 1 48 =SUMMENPRODUKT(A1:B1*B2:C2*A3:B3*B4:C4)
2 2 2 48 =A1*B2*A3*B4 + B1*C2*B3*C4
3 3 3
4 4 4
5 5 5  

 
Okay, schauen Sie noch ein zweites Mal hin. Es ist kaum anders als das Beispiel zuvor, lediglich die 
berechneten Zellen verlaufen versetzt durch die Matrix. Gut, das wissen wir nun.  

Matrix 3 

1
2
3
4
5

A B C D E F
1 1 1 30 =SUMMENPRODUKT(B1:B5*2)
2 2 2 30 =2*B1+2*B2+2*B3+2*B4+2*B5
3 3 3
4 4 4
5 5 5  

 
Das ist wieder leicht, nicht wahr? – Wir können also auch mit Konstanten rechnen.  

Matrix 4 

1
2
3
4
5

A B C D E F
1 1 1 48 =SUMMENPRODUKT(A2:B2*C3:C5)
2 2 2 48 =A2*C3 + A2*C4 + A2*C5 + B2*C3 + B2*C4 + B2*C5
3 3 3
4 4 4
5 5 5  

 
Haben Sie jetzt das erwartet? Die Zellen verlaufen einmal waagerecht und dann senkrecht! Und wir 
haben nicht einmal die gleiche Anzahl Zellen!!!  
 
Die erste Matrix verläuft waagerecht (A2:B2), die zweite senkrecht (C3:C5). Stellen Sie sich die 
Zwischenergebnisse in einer Matrix vor: zwei Spalten (A:B) und drei Zeilen (3, 4, 5) mit den 
jeweiligen Berechnungen sowie anschließender Summenbildung.  

 

1
2
3
4
5

A B C D E F

6 6 48 =SUMME(A1:C5)
8 8

10 10

 

Matrix 5 

1
2
3
4
5

A B C D E F
1 1 1 #NV =SUMMENPRODUKT(B1:B4*C1:C3)
2 2 2  =B1*C1+B2*C2+B3*C3+B4*???
3 3 3
4 4 4
5 5 5  

 
Nanu?! Im Beispiel zuvor haben wir gesehen, dass der SUMMENPRODUKT-Funktion egal zu sein 
scheint, wie die Zellen verlaufen – und jetzt das! Aber hat es nicht auch etwas Versöhnliches, dass 
wir hier nun endlich eine Fehlermeldung sehen?  
 
Worin unterscheiden sich die Beispiele 4 und 5?  
 
In diesem Beispiel (5) verlaufen beide Matrizen in die gleiche Richtung (vertikal), in Beispiel 4 
verläuft die eine Matrix horizontal, die andere vertikal. – Wenn Sie sich die Zwischenergebnisse 
wieder in einer Matrix vorstellen wollen, dann fehlt Ihnen einen Wert für die Multiplikation mit der 4 
aus Zelle B4. 
 

1
2
3
4
5

A B C D E F
1
4 #NV
9

 4*?

Hinweis für fortgeschrittene Anwender: Wenn Sie die Datentabelle aus Beispiel 5 analog des Beispiels 4 
berechnen wollen, nutzen Sie die Funktion MTRANS: 
=SUMMENPRODUKT(B1:B4*MTRANS(C1:C3)) 
 

Was zeigen uns die vorangegangenen Beispiel-Matrizen? 

SUMMENPRODUKT kann eine Konstante mit jeder ein- und mehrspaltigen Matrix berechnen (gleiches gilt 
für ein- und mehrzeilige Matrizen)  

=SUMMENPRODUKT(2*A1:C5) 
 
SUMMENPRODUKT kann eine einspaltige Matrix mit jeder einzeiligen Matrix berechnen; auf die 
Zellenanzahl kommt es hierbei nicht an.  

=SUMMENPRODUKT(A1:A5*B3:F3) 
 
In allen anderen Fällen, muss die Zellenanzahl gleich groß sein:  

• eine einspaltige Matrix mit einer ein- und mehrspaltigen Matrix  
• eine einzeilige mit einer ein- und mehrzeiligen Matrix  
• gemischte Formen aus mehrspaltigen mit mehrzeiligen Matrizen  

=SUMMENPRODUKT(A1:B5*A1:C5) 
 
Es gilt: Wenn Matrizen in die gleiche Richtung (vertikal / horizontal) verlaufen, dann müssen sie immer gleich 
groß sein.  
 
Lassen Sie uns die Theorie zu SUMMENPRODUKT an dieser Stelle abbrechen. Wir haben noch andere 
Aspekte zu betrachten…  
 

Wahrheitswerte /  bool’sche Werte  

Zwar wissen wir aus dem täglichen Leben, dass eine Frage nicht stets mit einem Ja oder Nein beantwortet 
werden kann. In Excel müssen wir allerdings stets darauf achten, Fragen so zu stellen, dass die Antwort nur 
Ja oder Nein lauten kann – andernfalls drohen unsere Ergebnisse fehlerhaft zu sein. Das kennen wir 
insbesondere von der Funktion WENN.  
 

=WENN ( Bedingung X erfüllt? ; dann ; sonst )  
Das „Dann“ folgt, wenn die Antwort „Ja“ lautet, das „Sonst“ folgt, wenn die Antwort „Nein“ lautet.  

 
Wahrheitswerte (auch bool’sche Werte genannt) sind die Ergebnisse WAHR und FALSCH. Im Sinne unserer 
vorstehenden Ausführungen steht das WAHR für ein „Ja“ und das FALSCH für ein „Nein“.  
 
Schauen wir uns folgende Beispiele an:  
 

1
2
3
4
5
6
7
8
9
10
11

A B C D E F
1 1 1 ja =WENN(B1=1;"ja";"nein")
2 2 2 nein =WENN(B1=2;"ja";"nein")
3 3 3
4 4 4 ja =WENN(B1=1;"ja")
5 5 5 FALSCH =WENN(B1=2;"ja")

WAHR =B1=1
FALSCH =B1=2

1 =(B1=1)*1
0 =(B1=2)*1  

 
E1/E2:  
Hier haben wir eine klassische WENN-Funktion, die auf die Frage (Bedingung) „Entspricht B1 exakt 
den Wert 1?“ mit den vorgegebenen Antworten „ja“ bzw. „nein“ (für den Wert 2) reagiert.  

E4/E5:  
Hier haben wir in der WENN-Funktion keine Definition für den Fall getroffen, dass die Bedingung 
nicht erfüllt ist (es fehlt das „sonst“-Argument). Ergebnis ist ein FALSCH in E5.  
 
E7/E8:  
Hier haben wir gänzlich auf die WENN-Funktion verzichtet. Excel antwortet hier auf die Behauptung 
„B1 entspricht exakt dem Wert 1“ mit WAHR und für den Wert 2 mit FALSCH.  
 
E10/E11:  
Dass wir Wahrheitswerte durch Multiplikation mit 1 in Zahlen wandeln können, haben Sie in unseren 
anderen Erläuterungen zu Funktionen bereits erfahren.  

 
Für uns von ganz besonderem Interesse sind die Wahrheitswerte, oben die Zellen E7 und E8. Denn mit der 
Kenntnis, ob eine Behauptung WAHR oder FALSCH ist, lässt sich sehr viel anfangen! 
 
 

Zählen und Summieren mit mehr als einer Bedingung 

ZÄHLENWENN kann mit einer Bedingung zählen – nur mit einer Bedingung! SUMMEWENN kann nur mit 
einer Bedingung summieren – nur mit einer Bedingung! Für das Zählen und Summieren bei mehr als einer 
Bedingung bietet sich SUMMENPRODUKT an:  
 

1
2
3
4
5
6

A B C D E F
Name Ort Zahl
Anton Berlin 10 3 =ZÄHLENWENN(A:A;"Anton")
Berta Düsseldorf 20 3 =SUMMENPRODUKT((A2:A6="Anton")*1)
Cäsar Köln 30 2 =SUMMENPRODUKT((A2:A6="Anton")*(B2:B6="Berlin"))
Anton Berlin 40 1 =SUMMENPRODUKT((A2:A6="Anton")*(B2:B6="Berlin")*(C2:C6=10))
Anton Wuppertal 50 50 =SUMMENPRODUKT((A2:A6="Anton")*(B2:B6="Berlin")*(C2:C6))  

 

Zählen mir einer Bedingung 

Zu E2 obigen Beispiels: 
 
Für das Zählen mit einer Bedingung greifen wir zu ZÄHLENWENN.  
 

Zu E3 obigen Beispiels:  
 
Dies wäre das SUMMENPRODUKT-Äquivalent zu der ZÄHLENWENN-Formel in E2. Auf die Frage: 
Enthält die Zelle A2 / A3 / A4 / A5 / A6 einen „Anton“? liefert SUMMENPRODUKT eine Matrix aus 
WAHR und FALSCH: 
 

1
2
3
4
5
6
7
8

A B C D
Name

WAHR 1 =A2*1
FALSCH 0 =A3*1
FALSCH 0 =A4*1
WAHR 1 =A5*1
WAHR 1 =A6*1

0 =SUMME(A2:A6) 3 =SUMME(C2:C6)  
 
Mit WAHR und FALSCH lässt sich rechnen – aber es muss auch getan werden (das Rechnen)! In 
A8 ist erkennbar, dass sich WAHR ohne einen Rechenschritt nicht summieren lässt – das Ergebnis 
ist eine unbefriedigende Null. Das ist nicht, was wir suchen.  
 
In C2:C6 wird nun gerechnet (mit 1 multipliziert) – dadurch wird aus WAHR eine 1 und aus FALSCH 
eine 0, also Zahlen, die sich – wie C8 zeigt – summieren lassen.  

Zählen mit zwei und mehr Bedingungen 

Zu E4 obigen Beispiels:  
 

Es gilt das gleiche Schema wie zuvor, nur haben wir nun zwei Bedingungen. Die Matrix der 
Wahrheitswerte sieht nun so aus:  

 

1
2
3
4
5
6
7
8

A B C D E
Name Ort

WAHR WAHR 1 =A2*B2
FALSCH FALSCH 0 =A3*B3
FALSCH FALSCH 0 =A4*B4
WAHR WAHR 1 =A5*B5
WAHR FALSCH 0 =A6*B6

2 =SUMME(D2:D7)  
 
Eine Multiplikation mit 1 ist entbehrlich, denn es wird gerechnet, so dass ein WAHR als 1 und ein 
FALSCH als 0 gelten. Die Multiplikation von 1*1 ergibt 1, die von 0*0 ergibt 0 und auch 0*1 und 1*0 
ergeben 0. Die 1 entsteht als Produkt nur, wenn sowohl der jeweilige Zeilenwert aus Spalte A und 
Spalte B den Wert WAHR enthalten.  
 

Zu E5 obigen Beispiels:  
 

Das nächste Ergebnis wird nicht überraschen und bedarf gewiss keiner weiteren Erläuterungen:  
 

1
2
3
4
5
6
7
8

A B C D E F
Name Ort Zahl

WAHR WAHR WAHR 1 =A2*B2*C2
FALSCH FALSCH FALSCH 0 =A3*B3*C3
FALSCH FALSCH FALSCH 0 =A4*B4*C4
WAHR WAHR FALSCH 0 =A5*B5*C5
WAHR FALSCH FALSCH 0 =A6*B6*C6

1 =SUMME(E2:E7)  
 

Summieren mit zwei und mehr Bedingungen 

Zu E6 obigen Beispiels. 
 

Im letzten Fall des Beispiels wird nun nicht mehr gezählt, sondern tatsächlich summiert: Die Fragen 
lauten:  

(1) Welche Zellen der Spalte A enthalten den String „Anton“?  
(2) Welche Zellen der Spalte B enthalten den String „Berlin“? 

 
Für Spalte C war nun keine Prüfung mehr vorgesehen – vielmehr sind die Inhalte (Zahlen) selbst zu 
berücksichtigen.  
 

1
2
3
4
5
6
7
8

A B C D E F
Name Ort Zahl

WAHR WAHR 10 10 =A2*B2*C2
FALSCH FALSCH 20 0 =A3*B3*C3
FALSCH FALSCH 30 0 =A4*B4*C4
WAHR WAHR 40 40 =A5*B5*C5
WAHR FALSCH 50 0 =A6*B6*C6

50 =SUMME(E2:E7)  
 
Die Wahrheitswerte der Spalten A und B sowie die tatsächlichen Werte aus Spalte C werden 
miteinander multipliziert und deren Ergebnisse (Produkte) werden summiert.  

Semikolon oder Multiplikation? 

Folgende Tabelle enthält versehentlich in C6 keine Zahl, sondern einen Text. Und diese Zelle C6 ist 
Bestandteil der in E6 und E7 vorgesehenen Berechnung: 
 

1
2
3
4
5
6
7

A B C D E F
Name Ort Zahl
Anton Berlin 10 3 =ZÄHLENWENN(A:A;"Anton")
Berta Düsseldorf 20 3 =SUMMENPRODUKT((A2:A6="Anton")*1)
Cäsar Köln 30 2 =SUMMENPRODUKT((A2:A6="Anton")*(B2:B6="Berlin"))
Anton Berlin 40 1 =SUMMENPRODUKT((A2:A6="Anton")*(B2:B6="Berlin")*(C2:C6=10))
Anton Wuppertal Hallo #WERT! =SUMMENPRODUKT((A2:A6="Anton")*(B2:B6="Berlin")*C2:C6)

50 =SUMMENPRODUKT((A2:A6="Anton")*(B2:B6="Berlin");C2:C6)  
 
Zu E6:  

 
Die SUMMENPRODUKT-Formel liefert den Fehler #WERT!  
 
Warum?  
 
Eine Multiplikation der Wahrheitswerte aus den Berechnungen (A2:A6=“Anton“) und (B2:B6=“Berlin“) 
mit diesem „Hallo“ in C6 ist nicht möglich. Denn „Hallo“ ist keine Zahl.  
 

Schreiben Sie in Zelle A1 den Text „Hallo“ und in Zelle A2 die Formel =A1*1  
Ergebnis ist die Fehlermeldung #WERT! 
 
Schreiben Sie nun in Zelle A3 die Formel =SUMME(A2) 
Ergebnis ist wiederum die Fehlermeldung #WERT! 

 
Erinnern Sie sich an unsere eingangs ausgeführten Erläuterungen zu einer oder mehreren 
Ergebnismatrizen? Genau das ist hier das Problem! Wir haben nur eine Matrix, und diese verursacht 
in einer Zelle (hier C6) einen Fehlerwert:  

 
Wir haben bestimmt, dass die Ergebnismatrix einspaltig sein soll und zwar fünf (Zellen 2, 3, 
4, 5, 6) Argumente umfassen soll. Das letzte Argument aber enthält hier den Fehlerwert.  

 
Zu E7:  

 
In dieser Version (mit dem Semikolon) toleriert SUMMENPRODUKT die vorzunehmende Addition.  
 
Warum?  
 
Wir haben hier zwei Ergebnismatrizen! Die erste enthält die Produkte aus den Wahrheitswerten von 
„Anton“ und „Berlin“ – das sind schlichte Einsen (1) und Nullen (0). Die zweite Matrix entspricht den 
Werten aus den Zellen C2:C6. Wobei nun für C6 das Übliche gilt: Text entspricht dem Wert Null (0).  
 

Schreiben Sie in Zelle A1 den Text „Hallo“ und in Zelle A2 die Formel =SUMME(A1)  
Ergebnis ist nun eine ordentliche Null (0).  

 
Hier haben wir folgende Situation: 
 

Wir haben bestimmt, dass zwei Ergebnismatrizen vorliegen sollen, die jeweils fünf 
Argumente umfassen sollen. Und das ist hier nun gegeben:  
 

Die erste Matrix enthält die Werte  { 1; 0; 0; 1; 1 }  
die zweite Matrix enthält die Werte  { 10; 20; 30; 40; „Hallo“ }.  

 
Die Matrizen sind hier also – im Gegensatz zur Version in E6 – mit Inhalten gefüllt. Und weil 
Text, wie wir wissen, dem Wert „Null“ entspricht, liefert uns SUMMENPRODUKT in dieser 
Schreibweise (mit Semikolon) keinen Fehlerwert.  

ODER-Bedingungen 

Unsere bisherigen Erläuterungen und Beispiele betrafen UND-Bedingungen. In folgender Formel summieren 
wir die Zellen des Bereichs C2:C6, deren Zeilenentsprechungen in Spalte A einen „Anton“ und in Spalte B 
ein „Berlin“ enthalten:  
 
=SUMMENPRODUKT((A2:A6=“Anton“)*(B2:B6=“Berlin“)*C2:C6) 
 
Es gibt auch ein ODER:  
 

1
2
3
4
5
6
7
8
9
10

A B C D E F
Name Ort Zahl
Anton Berlin 1 2 =SUMMENPRODUKT((A2:A10="Anton")*(B2:B10="Berlin")*C2:C10)
Berta Berlin 10 20 =SUMMENPRODUKT((A2:A10="Berta")*(B2:B10="Berlin")*C2:C10)
Cäsar Köln 100
Dieter Hamburg 200 22 =SUMMENPRODUKT(((A2:A10="Anton")+(A2:A10="Berta"))*(B2:B10="Berlin")*C2:C10)
Anton Berlin 1
Berta Berlin 10
Anton Wuppertal 1
Berta Wuppertal 10
Egon Wuppertal 300  

 
E2 / E3:  

 
Das kennen Sie bereits: Wir summieren die Werte aus Spalte C für alle „Anton“ aus „Berlin“ bzw. alle 
„Berta“ aus „Berlin“.  
 

E5:  
 
Hier haben wir nun eine ODER-Variante:  
 
Für Zellen A2:A10 wird gefragt, ob sie den „Anton“ enthalten (falls ja bekommen wir ein WAHR, 
sonst ein FALSCH). In Matrixform sieht diese Zwischenergebnis so aus:  

{ WAHR; FALSCH; FALSCH; FALSCH; WAHR; FALSCH; WAHR; FALSCH; FALSCH } 
 
Für dieselben Zellen wird weiter gefragt, ob sie die „Berta“ enthalten (auch hier bekommen wir ein 
WAHR oder ein FALSCH): 

{ FALSCH; WAHR; FALSCH; FALSCH; FALSCH; WAHR; FALSCH; WAHR; FALSCH } 
 
Beachten Sie die Klammern in der Formel, das ist wichtig! Wir wissen, dass bei jeder Berechnung 
ein WAHR als 1 und ein FALSCH als 0 gilt. Wenn wir nun also die bool’schen Werten obiger 
Matrizen addieren, erhalten wir:  
 { 1; 1; 0; 0; 1; 1; 1; 1; 0 } 
 
Schließlich fragen wir noch den Bereich B2:B10 nach einem „Berlin“ ab. Diese Ergebnismatrix sieht 
so aus:  

{ WAHR; WAHR; FALSCH; FALSCH; WAHR; WAHR; FALSCH; FALSCH; FALSCH } 
 
Multipliziert mit unserer vorherigen Matrix aus Einsen und Nullen haben wir nun folgendes Ergebnis: 
 { 1; 1; 0; 0; 1; 1; 0; 0; 0 } 
 
Abschließend haben wir noch die Matrix C2:10 mit den Werten 
 { 1; 10; 100; 200; 1; 10; 1; 10; 300 } 
 
Und nun noch die Multiplikation dieser letzten zwei Matrizen:  
 { 1; 10; 0; 0; 1; 10; 0; 0; 0 } 
 
Noch die Summe aller Matrixwerte = 22.  

Beispielanwendungen 

Die Nutzung der Funktion SUMMENPRODUKT ist sehr vielseitig, weil sie die Prüfung mehrerer 
Bedingungen ermöglicht. Mit folgenden Beispielen wollen wir Ihnen einige wenige Möglichkeiten aufzeigen.   
 

Beispiel  1 

1
2
3
4
5
6
7
8

A B C D E F G H
Datum Nr. Wert Nr 1-Jan-07 2-Jan-07 3-Jan-07

1-Jan-07 123 15 122 50 0 0
1-Jan-07 123 10 123 25 7 0
2-Jan-07 123 7 124 0 150 100
1-Jan-07 122 50
3-Jan-07 124 100
2-Jan-07 124 150

 
 
Die Basisdaten sehen Sie in Spalten A bis C: Eine Liste mit Datumswerten und anderen Zahlen (hier z. B. 
Artikelnummern und Umsatzwerte). Ziel ist die Darstellung wie in Spalte E bis H: Es sollen Summen je 
Artikelnummer und Tag gebildet in dieser Darstellung gebildet werden.  
 
Die Formel in F2 (kopierbar bis H4) lautet:  
=SUMMENPRODUKT(($B$2:$B$10=$E2)*($A$2:$A$10=F$1)*$C$2:$C$10) 
 

Beispiel  2 

1
2
3
4
5
6
7

A B C D E F G H
Datum Nr. Wert Nr 1 2

1-Jan-07 123 15 122 0 0 50
2-Feb-07 123 10 123 22 10 0
7-Jan-07 123 7 124 0 100 150
3-Mrz-07 122 50
4-Feb-07 124 100
7-Mrz-07 124 150

3

 
 
In Fortführung des Beispiels 1 hier eine Kumulation für Monate. Die Zahlen in F1 (1), G1 (2) und H1 (3) 
stehen für die Monate (Januar, Februar, März).  
 
Die Formel in F2 (kopierbar bis H4) lautet: 
=SUMMENPRODUKT(($B$2:$B$10=$E2)*(MONAT($A$2:$A$10)=F$1)*$C$2:$C$10) 

Beispiel  3 

1
2
3
4
5
6
7
8
9

10
11
12
13
14
15

A B C D E F G H I J K

KW Tag Umsatz KW Mo Di Mi Do Fr Sa
2 Mo 10 1 0 0 0 0 0 0
2 Di 11 2 10 11 12 13 14 0
2 Mi 12 3 101 15 10 11 17
2 Do 13 4 5 120 0 0 0 0
2 Fr 14 5 0 0 0 0 0 0
3 Mo 101
3 Di 15
3 Mi 10
3 Do 11
3 Fr 17
4 Mo 5
4 Di 120

Umsatzliste - Ausgangsbasis Umsatzliste - andere Darstellung

0

 
 
Die Formel in F2 (kopierbar bis H4) lautet: 
=SUMMENPRODUKT(($A$2:$A$101=$E3)*($B$2:$B$101=F$2);$C$2:$C$101) 
 

Beispiel  4 

1
2
3
4
5
6
7
8
9

10
11

A B C D E F G H
PLZ Straße Anzahl PLZ Ort Straße + Nr.

12345 Theodor-Heuss-Str 1 12345 Musterstadt Theodor-Heuss-Str. 36
26129 Theodor-Heuss-Str 3 12345 Musterstadt Friedenstr. 7

12345 Musterstadt Friedenplatz 18

12345 Frieden 2 ! 

26129 4
44444 Beowulfsweg 0 26129 Oldenburg Theodor-Heuss-Str. 28

11 26129 Oldenburg   Theodor-Heuss-Str. 34  << enthält vorn Leerzeichen!
26129 Oldenburg Theodor-Heuss-Straße 36
26129 Oldenburg Bloherfelder Str. 117
26130 Oldenburg ABC-Str. 3
26130 Oldenburg ABC-Str. 4

 
 
Aufgabenstellung ist hier die Zählung in Spalte C - die Formel in C2 (nach unten kopierbar) lautet: 
=SUMMENPRODUKT((E$2:E$21=A2)*(LINKS(GLÄTTEN(G$2:G$21);LÄNGE(B2))=B2)) 
 

Beispiel  5 

1
2
3
4
5
6
7

A B C D E F G H I
Datum KW Artikel Wert 1 2

02.01.07 1 122 15 EUR 122 15 100 0
04.01.07 1 123 7 EUR 123 7 50 0
05.01.07 1 124 10 EUR 124 10 0 150
08.01.07 2 122 100 EUR
10.01.07 2 123 50 EUR
15.01.07 3 124 150 EUR

3

 
 
Aufgabenstellung ist eine Summierung der als Text vorliegenden Werte in Zellen D2:D7 nach 
Kalenderwochen – Darstellung: Spalte F bis I.  
=SUMMENPRODUKT(($B$2:$B$7=G$1)*($C$2:$C$7=$F2)*WECHSELN($D$2:$D$7;"EUR";"")) 

